

Kurzeinführung in die Fehlerrechnung für das Physikpraktikum

Mit Mathematica Code Beispielen

Julien Kluge - 7. Juni 2022 - Version 2.0.0

Inhaltsverzeichnis

1	Mittelwerte	2
1.1	Arithmetischer Mittelwert	2
1.2	Gewichteter Mittelwert	2
2	Pythagoreische Addition	5
3	Standardabweichung	6
4	Vertrauensbereich / zufällige Messunsicherheit	8
4.1	Student T-Wert	8
4.2	Größtfehlerabschätzung	10
5	Runden und signifikante Stellen	11
5.1	DIN	12
5.2	Angabe von Naturkonstanten	12
6	Kochrezept: Messwert berechnen	12
7	Fehlerfortpflanzung	14
7.1	Unkorrelierte Fortpflanzung	14
7.2	Exkurs: Korrelation	16
7.3	Korrelierte Fortpflanzung	18
8	Regressionen	20
8.1	Fit-Modell und Linearität	20
8.2	Gewichtete Regression	22
8.3	R^2 -Test	23
8.4	χ^2/dof -Test	24
8.5	Kovarianz- und Korrelationsmatrix	25
8.6	Korrelationen linearer Funktionen	26

1 Mittelwerte

1.1 Arithmetischer Mittelwert

Der *normale*, arithmetische Mittelwert \bar{x} sollte noch aus der Schule bekannt sein. Man addiert einfach alle Werte x_i zusammen, und teilt durch die Anzahl n aller Werte.

$$\bar{x} = \frac{x_1 + x_2 + \cdots + x_{n-1} + x_n}{n} \quad (1)$$

Mit Summenschreibweise (die wir als Physiker natürlich bevorzugen weil sie kompakter ist) ergibt sich die Standardmäßige Definition:

Definition 1: Arithmetischer Mittelwert

$$\bar{x} = \frac{1}{n} \sum_i^n x_i \quad (2)$$

\bar{x} Arithmetischer Mittelwert
 n Anzahl aller (Mess)Werte
 x_i i -ter (Mess)Wert

Mathematica Code 1: Arithmetischer Mittelwert

Wir definieren eine Liste aller (Mess)Werte und berechnen den Mittelwert mit dem Befehl **Mean**:

```
messwerte={1.0456,0.9774,1.0023,0.9904,1.3995};  
Mean[messwerte]
```

```
1.08304
```

1.2 Gewichteter Mittelwert

Man stelle sich folgende Situation vor: man hat die Werte x_i und möchte aus denen ein Mittelwert bilden. Jetzt weiß man aber, dass einige dieser Werte viel genauer und/oder weniger unsicher sind als andere. Da hätte man es doch lieber, diese Werte stärker in den Mittelwert eingehen zu lassen. Dafür existiert der gewichtete Mittelwert. In mathematischen Worten also Folgendes: man habe die Werte x_i mit den Unsicherheiten u_i und führe einen Gewichtungsfaktor p_i ein, der von der Unsicherheit abhängt: $p_i = p_i(u_i)$. Jetzt ist uns klar, dass wenn die Unsicherheit u_i klein ist, dann wollen wir den Wert stärker gewichten, andersherum wollen wir eine kleine Gewichtung für größere Unsicherheiten. Des Weiteren gilt, die statistische Streuung eines Wert mit Unsicherheit u_i ist einfach mit dem Quadrat gegeben, es gilt also:

$$p_i \propto \frac{1}{u_i^2} \quad (3)$$

Unter Einführung einer Proportionalitätskonstante C bekommen wir

$$p_i = \frac{C}{u_i^2} \quad (4)$$

Wenn man diesen Gewichtungsfaktor in die Formel (6) und Formel (7) einsetzt, bemerkt man dass der Proportionalitätsfaktor fast frei gewählt werden darf mit $C \in \mathbb{R} \setminus \{0\}$. Für alle, die mit einer Programmiersprache arbeiten, kann der Faktor demnach der Einfachheit halber zu eins gesetzt werden:

Definition 2: Instrumenteller Gewichtungsfaktor

$$p_i = \frac{1}{u_i^2} \quad (5)$$

p_i Gewichtungsfaktor des Werts $(x_i \pm u_i)$

u_i Unsicherheit des Werts $(x_i \pm u_i)$

Für alle anderen, die per Hand mit Taschenrechner, Excel, o.Ä. rechnen, kann eine Vereinfachung gemacht werden indem man $C = u_1^2$ setzt. Damit wird $p_1 \equiv 1$ und alle anderen Gewichtungsfaktoren sind relative Quotienten zu p_1 . Das ist aber natürlich nicht unbedingt notwendig. Jetzt kann das gewichtete Mittel definiert werden mit:

Definition 3: Gewichteter Mittelwert

$$\bar{x} = \frac{\sum_i^n p_i \cdot x_i}{\sum_i^n p_i} \quad (6)$$

\bar{x} Gewichtungsfaktor des Werts $(x_i \pm u_i)$

n Anzahl aller Werte

x_i i -ter Wert

p_i i -ter Gewichtungsfaktor

Natürlich ist auch die Unsicherheit des Mittelwertes damit betroffen:

$$\bar{u} = \pm \frac{\sqrt{\sum_i^n (p_i \cdot u_i)^2}}{\sum_i^n p_i} \quad (7)$$

Durch Einsetzen des instrumentellen Gewichtungsfaktors und Vereinfachung erhält man:

Definition 4: Unsicherheit des gewichteten Mittelwerts

$$\bar{u} = \pm \sqrt{\frac{C}{\sum_i^n p_i}} \quad \xrightarrow{C=1} \quad \bar{u} = \pm \left(\sum_i^n p_i \right)^{-1/2} \quad (8)$$

\bar{u} Unsicherheit des gewichteten Mittelwerts

n Anzahl aller Werte

C frei wählbarer Proportionalitätsfaktor

p_i i -ter Gewichtungsfaktor

Beispiel 1: Gewichteter Mittelwert

Folgende Werte der Erdbeschleunigung g mit Unsicherheit u wurden gemessen:

$g \text{ [m/s]}$	9.81	9.79	9.80	9.60
$\pm u \text{ [m/s]}$	0.03	0.11	0.04	0.70

Mit $C = 1$ berechnet sich:

$p = 1/u^2$	$p \cdot g$
1111.11	10900
82.6446	809.091
625	6125
2.04082	19.5918
Σ	1820.8 17853.7

Es folgt das Ergebnis mit:

$$\bar{g} = \frac{\sum_i^n p_i \cdot g_i}{\sum_i^n p_i} = \frac{17853.7}{1820.8} = 9.80542 \quad (9)$$

$$\bar{u} = \pm \frac{1}{\sqrt{\sum_i^n p_i}} = \pm \frac{1}{\sqrt{1820.8}} = \pm 0.02344 \quad (10)$$

Und damit:

$$g = (9.81 \pm 0.03) \text{ m/s} \quad (11)$$

Mathematica Code 2: Gewichteter Mittelwert

Wir können entweder die Definition direkt anwenden (**Sqrt**: Quadratwurzel, **Total**: Summe einer Liste):

```
g={9.81,9.79,9.80,9.60};
u={0.03,0.11,0.04,0.70};
p=1/u^2;
Total[g*p]/Total[p]
1/Sqrt[Total[p]]
```

```
9.80542
0.023435
```

Oder man kann über den Unsicherheitsbefehl **Around** direkt den gewichteten Mittelwert mit **MeanAround** bestimmen:

```
g={Around[9.81,0.03],Around[9.79,0.11],
  Around[9.80,0.04],Around[9.60,0.70]};
MeanAround[g]
```

```
9.805±0.023
```

Wichtige Anmerkung 1

Das gewichtete Mittel darf nur in Fällen gebildet werden, in denen sicher ist, dass die Werte normalverteilt sind. Im Allgemeinen kann man sagen: bei zu vielen Ausreißern sollte man auf das normale arithmetische Mittel zurückgreifen.

2 Pythagoreische Addition

Wenn man mehrere Unsicherheiten für einen Wert zusammenführen will, muss das geometrisch geschehen. Dafür benutzt man die pythagoreische Addition. Sie ist folgendermaßen definiert:

$$u = \sqrt{u_1^2 + u_2^2 + \cdots + u_{n-1}^2 + u_n^2} \quad (12)$$

Mit Summenschreibweise führt das auf:

Definition 5: Pythagoreische Addition

$$u = \sqrt{\sum_i^n x_i^2} \quad (13)$$

u Pythagoreisch addierter Wert
 n Anzahl aller Werte
 u_i i -ter Wert

3 Standardabweichung




Abbildung 1: Normalverteilung mit Mittelwert μ und Standardabweichung σ .

Ist ein Messwert mit einem zufälligen Fehler belegt, so streuen diese um einen gemeinsamen Punkt (ohne systematische Abweichungen ist dieser Punkt der Wahrheitswert). Dabei nimmt diese Streuung die Form einer Gaußschen Normalverteilung $N(x, \mu, \sigma)$ (auch Glockenkurve genannt) an. Die Form einer Glocke erklärt sich leicht, wenn man annimmt, dass Streuungen mit steigendem Abstand vom Mittelwert μ zunehmend unwahrscheinlicher werden. Die Wahrscheinlichkeit selbst ist die Fläche unter der Kurve. Es gilt also:

$$\int_{-\infty}^{\infty} dx N(x, \mu, \sigma) = 1 \quad (14)$$

Man sieht außerdem leicht, dass diese Verteilung symmetrisch um den Mittelwert μ ist. Damit können nun symmetrische Intervalle festgelegt werden, in denen eine bekannte Wahrscheinlichkeit existiert, dass eine Messung x' dort landet. Diese Intervalle nennt man Standardabweichung σ und sind selbst Parameter der Normalverteilung. Für die ersten fünf Intervalle gilt:

Intervall	statistische Sicherheit
$\mu \pm 1\sigma$	$\approx 68.26895\%$
$\mu \pm 2\sigma$	$\approx 95.44997\%$
$\mu \pm 3\sigma$	$\approx 99.73002\%$
$\mu \pm 4\sigma$	$\approx 99.99367\%$
$\mu \pm 5\sigma$	$\approx 99.99994\%$

Wir rechnen in der Physik für gewöhnlich mit einer statistischen Unsicherheit von einer Standardabweichung (ein Sigma). Hat man eine Messreihe gemacht, kann die Standardabweichung davon berechnet werden mit:

Definition 6: Empirische Standardabweichung

$$\sigma = \sqrt{\frac{1}{n-1} \sum_i^n (x_i - \mu)^2} \quad (15)$$

σ Standardabweichung (auch häufig s)
 n Anzahl aller Messwerte
 x_i i -ter Messwert
 μ Arithmetischer Mittelwert \bar{x} der Messwerte

Beispiel 2: Standardabweichung

Folgende zehn Messwerte ($n = 10$) für die Erdbeschleunigung g [m/s] wurden aufgenommen:

9.81279, 9.83616, 9.76557, 9.78496, 9.84230
 9.75096, 9.72767, 9.84866, 9.74724, 9.76847

Der Mittelwert ist

$$\bar{g} = \mu = 9.788478 \quad (16)$$

Wir berechnen

	$(g - \mu)^2$
	0.0005910733
	0.0022735731
	0.0005247765
	0.0000123763
	0.0028968077
	0.0014076003
	0.0036976129
	0.0036218731
	0.0017005726
	0.0004003201
Σ	0.0171265860

Und somit ergibt sich die Standardabweichung zu:

$$\sigma = \sqrt{\frac{1}{10-1} \cdot 0.0171265860} \quad (17)$$

$$\approx 0.0436228610 \quad (18)$$

Mathematica Code 3: Standardabweichung

Mathematica kennt einen fertigen Befehl **StandardDeviation** für die Standardabweichung:

```
g={9.81279,9.83616,9.76557,9.78496,9.84230,  
  9.75096,9.72767,9.84866,9.74724,9.76847};  
StandardDeviation[g]
```

```
0.043622861
```

4 Vertrauensbereich / zufällige Messunsicherheit

Mithilfe der Standardabweichung lässt sich, in einer bestimmten statistischen Wahrscheinlichkeit, die zufällige Messunsicherheit abschätzen. Diese Abschätzung nennt sich Vertrauensbereich und ist folgendermaßen definiert:

Definition 7: Vertrauensbereich

$$\bar{s} = \pm t \cdot \frac{\sigma}{\sqrt{n}} \quad (19)$$

\bar{s} Vertrauensbereich
 σ Standardabweichung
 n Anzahl aller Messwerte
 t Student T-Wert

4.1 Student T-Wert

Der Student-Faktor (benannt nach William Sealy Gosset der unter dem Pseudonym Student gearbeitet hat) kann auf komplizierte Weise, von der Student T-Verteilung berechnet werden. Diese Verteilung besitzt nur die Anzahl der Messwerte n (eigentlich Freiheitsgrade: $\nu = n - 1$) als Parameter und ist um null symmetrisch. In den Einführungspraktika und Grundpraktika I & II ist es erlaubt die Näherung $t \approx 1$ für $n \geq 6$ zu machen. Will man diese Näherung nicht eingehen, kann man in folgende Tabelle gucken und den Wert ablesen.

n	$\pm 1\sigma$	$\pm 2\sigma$	$\pm 3\sigma$	$\pm 4\sigma$	$\pm 5\sigma$
2	1.83734	13.9677	235.801	10050.4	1.1×10^6
3	1.32128	4.52654	19.2067	125.641	1320.71
4	1.19688	3.30682	9.21894	32.6164	156.678
5	1.14163	2.86931	6.62021	17.4482	56.8484
6	1.11051	2.64865	5.50708	12.2814	31.8470
7	1.09057	2.51652	4.90406	9.84416	22.0199
8	1.07671	2.42881	4.52997	8.46693	17.1025
9	1.06653	2.36642	4.27663	7.59506	14.2495
10	1.05873	2.31981	4.09426	6.99864	12.4220
11	1.05256	2.28368	3.95694	6.56718	11.1662
12	1.04757	2.25486	3.84994	6.24163	10.2571
13	1.04344	2.23135	3.76428	5.98780	9.57203
14	1.03997	2.21180	3.69419	5.78466	9.03909
15	1.03701	2.19529	3.63580	5.61858	8.61377
16	1.03446	2.18116	3.58642	5.48038	8.26710
17	1.03224	2.16894	3.54413	5.36364	7.97950
18	1.03029	2.15826	3.50750	5.26377	7.73733
19	1.02856	2.14885	3.47547	5.17739	7.53077
20	1.02702	2.14049	3.44723	5.10196	7.35262
21	1.02563	2.13303	3.42215	5.03554	7.19748
22	1.02438	2.12631	3.39973	4.97661	7.06121
23	1.02325	2.12024	3.37956	4.92398	6.94062
24	1.02222	2.11473	3.36132	4.87670	6.83316
25	1.02127	2.10970	3.34475	4.83400	6.73684
26	1.02040	2.10509	3.32963	4.79524	6.65001
27	1.01960	2.10085	3.31578	4.75990	6.57136
28	1.01886	2.09694	3.30305	4.72756	6.49979
29	1.01818	2.09333	3.29130	4.69785	6.43440
30	1.01754	2.08997	3.28043	4.67045	6.37442
31	1.01695	2.08684	3.27033	4.64512	6.31921
32	1.01639	2.08393	3.26094	4.62163	6.26824
33	1.01587	2.08120	3.25218	4.59978	6.22102
34	1.01538	2.07865	3.24399	4.57942	6.17717
35	1.01492	2.07625	3.23631	4.56038	6.13635
36	1.01449	2.07400	3.22910	4.54255	6.09824
37	1.01408	2.07187	3.22231	4.52582	6.06258
38	1.01370	2.06986	3.21592	4.51009	6.02916
39	1.01333	2.06796	3.20988	4.49527	5.99777
40	1.01299	2.06617	3.20417	4.48128	5.96822
50	1.01031	2.05232	3.16051	4.37526	5.74678
100	1.00508	2.02557	3.07755	4.17845	5.34785
1000	1.00050	2.00251	3.00752	4.01708	5.03272

Wir sehen, dass die im Praktikum verwendete Näherung mit $n \geq 6$ willkürlich und den Versuchen zweckmäßig gewählt worden ist. Der Student-T Wert ist $t(n = 6) \approx 1.11051$ d.h. bei $n = 6$ unterschätzen wir den Vertrauensbereich/die zufällige Messunsicherheit um circa 11%.

4.2 Größtfehlerabschätzung

Es stellt sich die Frage, was man benutzt wenn $n < 6$ und die Näherung des Student T-Wertes nicht mehr gemacht werden kann. In diesem Fall kann man eine Größtfehlerabschätzung machen. In einfachen Worten nimmt man die größtmöglichen Abweichungen der Messreihe an, indem man sich anguckt welcher Wert am Weitesten vom Mittelwert entfernt ist und nimmt das als Unsicherheit. Es gilt also:

Definition 8: Größtfehlerabschätzung als Vertrauensbereich

$$\bar{s} \approx \pm \max_i |x_i - \bar{x}| \quad (20)$$

\bar{s} Abgeschätzter Vertrauensbereich
 x_i i -ter Messwert
 \bar{x} Mittelwert

Beispiel 3: Vertrauensbereich mit Standardabweichung

Wir nehmen die Werte des vorherigen Beispiels:

$$\begin{aligned} \sigma &\approx 0.04362 \\ n &= 10 \end{aligned}$$

Es ergibt sich **ohne Student T-Wert**:

$$\bar{s} = \pm \frac{0.04362}{\sqrt{10}} \approx \pm 0.01379 \quad (21)$$

Mit Student T-Wert ($t(n=10) \approx 1.05873$):

$$\bar{s} = \pm 1.05873 \cdot \frac{0.04362}{\sqrt{10}} \approx \pm 0.01460 \quad (22)$$

Mit **Größtfehlerabschätzung** (auch wenn sie hier wirklich nutzlos ist):

$$\begin{aligned} \bar{s} &= \pm \max(0.0243, 0.0477, 0.0229, 0.0035, 0.0538, 0.0375, 0.0608, 0.0602, 0.0412, 0.0200) \\ \bar{s} &= \pm 0.0608 \end{aligned} \quad (23)$$

Mathematica Code 4: Vertrauensbereich

Mathematica kann alle drei Varianten spielend ausrechnen (**Length**: Länge einer Liste):

```
g = {9.81279, 9.83616, 9.76557, 9.78496, 9.84230,
     9.75096, 9.72767, 9.84866, 9.74724, 9.76847};
```

```
StandardDeviation[g]/Sqrt[Length[g]]
```

```
InverseCDF[
  StudentTDistribution[Length[g] - 1],
  (1 + Erf[1/Sqrt[2]])/2
]*StandardDeviation[g]/Sqrt[Length[g]]
```

```
Max[Abs[g - Mean[g]]]
```

```
0.01379476
0.014604894
0.060808
```

Alternativ kann Mathematica bereits mit **MeanAround** die meiste Arbeit übernehmen:

```
g={9.81279,9.83616,9.76557,9.78496,9.84230,
   9.75096,9.72767,9.84866,9.74724,9.76847};
MeanAround[g]
```

```
9.788±0.014
```

5 Runden und signifikante Stellen

Ergebnisse auf 20 Stellen genau zu berechnen, ist zwar eine tolle Leistung, aber nicht sehr sinnvoll. Es macht keinen Sinn, Werte jenseits von ihren Fehlergrenzen genau anzugeben, deswegen müssen wir auf richtige Stellen runden. Im Allgemeinen runden wir so, dass die erste Stelle des Fehlers ungleich null, gleichzeitig die letzte Ziffer von rechts ist. Dabei wird der Messwert kaufmännisch gerundet, während der Fehler stets aufgerundet wird. Dabei müssen nachgehende Nullen vom Wert zwingend mitgeschrieben werden. Die so noch erhaltenen Stellen werden als signifikant bezeichnet.

Beispiel 4: Runden

$$(6.3279 \pm 0.057) \longrightarrow (6.33 \pm 0.06) \quad (24)$$

$$(36.03 \pm 0.41) \longrightarrow (36.0 \pm 0.5) \quad (25)$$

5.1 DIN

Wie für alles, existiert eine eigene DIN-Norm mit einer Eigenheit. Es gilt, wenn die Unsicherheit mit der ersten Stelle auf eins oder zwei geht, dann muss die Signifikanz auf eine weitere Stelle ausgedehnt werden. Nachgehende Nullen die hierbei durch Runden entstehen können, müssen dieses mal auch beim Fehler mitgenommen werden.

Beispiel 5: Runden nach DIN

$$(6.3279 \pm 0.134) \longrightarrow (6.33 \pm 0.14) \quad (26)$$

$$(36.003 \pm 0.148) \longrightarrow (36.00 \pm 0.15) \quad (27)$$

$$(15.437 \pm 0.297) \longrightarrow (15.44 \pm 0.30) \quad (28)$$

5.2 Angabe von Naturkonstanten

Bei der Angabe von Naturkonstanten kann man teilweise eine andere Notation von Unsicherheiten beobachten. Hier wird hinter dem Wert die Unsicherheit eingeklammert, welche mit signifikanter Stellenzahl von rechts gesehen geht.

Beispiel 6: Notation von Naturkonstanten

Das plancksche, reduzierte Wirkungsquantum wurde vor der Neudefinition 2019 definiert mit:

$$\hbar = 1.054571800(13) \times 10^{-34} \text{ J s} \quad (29)$$

$$= (1.054571800 \pm 0.000000013) \times 10^{-34} \text{ J s} \quad (30)$$

Der Vorteil dieser Notation ist offensichtlich. Einige Beispiele:

$$(6.33 \pm 0.06) \longrightarrow 6.33(6) \quad (31)$$

$$(36.0 \pm 2.5) \longrightarrow 36.0(2.5) \quad (32)$$

$$(6.33 \pm 0.14) \longrightarrow 6.33(14) \quad (33)$$

$$(15.44 \pm 0.30) \longrightarrow 15.44(30) \quad (34)$$

6 Kochrezept: Messwert berechnen

Hat man nun sukzessiv die erste Messreihe für einen Wert durchgeführt, gibt es ein Kochrezept wie wir diesen sowie seinen Fehler ordentlich berechnen können:

Definition 9: Messwertberechnung

1. Mittelwert berechnen

$$\bar{x} = \frac{1}{n} \sum_i^n x_i \quad (35)$$

2. Bekannte systematische Fehler korrigieren (sehr selten)

$$\bar{x}_k = \bar{x} - u_{\text{sys}} \quad (36)$$

3. Vertrauensbereich berechnen

$$\bar{s} = \pm \begin{cases} \frac{\sigma}{\sqrt{n}} = \sqrt{\frac{1}{n(n-1)} \sum_i^n (x_i - \bar{x})^2} & n \geq 6 \\ \max_i |x_i - \bar{x}| & n < 6 \end{cases} \quad (37)$$

4. Messunsicherheit durch pythagoreisches Addieren aller Unsicherheiten berechnen

$$u = \sqrt{\bar{s}^2 + u_{\text{gerät}}^2 + u_{\text{ablese}}^2 + \dots} \quad (38)$$

5. Ergebnis auf signifikante Stellen runden, mit Einheit (und Zehnerpotenz oder SI-Präfix) angeben.

$$(\bar{x}_k \pm u) \times 10^m [\text{SI-Präfix Einheit}] \quad (39)$$

Wichtige Anmerkung 2

Bei Angabe von Einheiten muss stets darauf geachtet werden, dass diese **nicht kursiv** geschrieben sind um sie nicht mit Variablen zu verwechseln.

Mathematica Code 5: Messwertberechnung

Dieses Beispiel enthält tatsächlich reale Daten. Macht daraus was ihr wollt.

Im Panthéon in Paris liegen neben berühmten Franzosen auch einige berühmte Physiker begraben. Ebenfalls hängt dort ein 67 m langes Foucaultsches Pendel in der Hauptkuppel herunter. Es wurden 17 Periodendauern mit einem Handy gemessen und zu einem Messwert berechnet. Dabei wurden angenommen: zweimal 0.15 s Unsicherheit in der Reaktionszeit, 10 ms Ableseungenauigkeit (1dgt.) und $10^{-4} \cdot t$ systematische Unsicherheit der Handyuhr.

```
tMess={16.38,16.66,16.54,16.38,16.31,16.23,16.56,16.52,16.32,
        16.48,16.41,16.32,16.38,16.34,16.27,16.35,16.44};
t=Mean[tMess] (*Mittelwert*)
statU=StandardDeviation[tMess]/Sqrt[Length[tMess]]; (*Vertrauensbereich*)
reaktU=0.15; (*Reaktionszeit*)
ableseU=0.01; (*Ableseungenauigkeit*)
systU=0.0001*t; (*syst. Unsicherheit Uhr*)
ut=Sqrt[statU^2+2*reaktU^2+ableseU^2+systU^2] (*Unsicherheit*)
```

16.4053
0.214174

Anstatt all diese Formeln per Hand zu benutzen, können und sollten wir die neuen Mathematica (Version 12.3 oder höher) Features der Fehlerrechnung benutzen:

```
tMess={16.38,16.66,16.54,16.38,16.31,16.23,16.56,16.52,16.32,
        16.48,16.41,16.32,16.38,16.34,16.27,16.35,16.44};
t=MeanAround[tMess]; (*Mittelwert und Vertrauensbereich*)
t=t+Around[0,0.15]+Around[0,0.15]; (*Reaktionszeit x2*)
t=t+Around[0,0.01]; (*Ableseungenauigkeit*)
t=t+Around[0,0.0001*t["Value"]] (*syst.Unsicherheit Uhr*)
```

16.41±0.21

7 Fehlerfortpflanzung

Wenn man eine Funktion $y = f(a, b, c, \dots)$ mit den Werten a, b, c, \dots und den dazugehörigen Unsicherheiten u_a, u_b, u_c, \dots hat, kann man leicht y ausrechnen. Doch wie rechnet man die Unsicherheit u_y für y aus? Dafür existiert die Gaußsche Fehlerfortpflanzung.

7.1 Unkorrelierte Fortpflanzung

Wenn die einzelnen Werte x_i mit den Unsicherheiten u_i nicht miteinander korreliert sind und die Funktion $f(x_1, x_2, \dots, x_n)$ ausgerechnet werden soll, dann kann man die Gaußsche Fehler-

fortpflanzung darstellen mit:

Definition 10: Unkorrelierte Gausßsche Fehlerfortpflanzung

$$u_f = \sqrt{\sum_i^n \left(\frac{\partial f(x_1, x_2, \dots, x_n)}{\partial x_i} \cdot u_i \right)^2} \quad (40)$$

u_f Fortgepflanzte Unsicherheit der berechneten Funktion $f(x_1, x_2, \dots, x_n)$
 n Anzahl aller Werte
 x_i i -te Variable/Wert
 u_i i -te Unsicherheit

Man kann sich diese Formel sogar ganz anschaulich darstellen. Die partielle Ableitung $\partial f / \partial x_i$ am Punkt (x_1, x_2, \dots, x_n) zeigt den Anstieg nach x_i und damit, wie empfindlich die Formel auf Änderung dieser Variable reagiert. Es fungiert somit als Gewichtung zur Unsicherheit u_i welche schlussendlich pythagoreisch mit allen anderen addiert wird.

Beispiel 7: Unkorrelierte Fehlerfortpflanzung

Nach Ohmschen Gesetz gilt folgender Zusammenhang zwischen Spannung U , Stromstärke I und elektrischer Widerstand R :

$$R = \frac{U}{I} \quad (41)$$

Man nehme jetzt folgende Werte (ungerundet) an:

$$U = (238.46 \pm 7.34) \text{ V} \quad (42)$$

$$I = (0.9239 \pm 0.0081) \text{ A} \quad (43)$$

Durch schnelle Rechnung ergibt sich:

$$R \approx 258.102 \Omega \quad (44)$$

Die Unsicherheit u_R von R bestimmt sich unter Vernachlässigung der Korrelation durch Fehlerfortpflanzung so:

$$u_R = \sqrt{\left(\frac{\partial R}{\partial U} \cdot u_U \right)^2 + \left(\frac{\partial R}{\partial I} \cdot u_I \right)^2} \quad (45)$$

$$= \sqrt{\left(\frac{1}{I} \cdot u_U \right)^2 + \left(-\frac{U}{I^2} \cdot u_I \right)^2} \quad (46)$$

$$= \sqrt{\left(\frac{1}{0.9239} \cdot 7.34 \right)^2 + \left(\frac{238.46}{(0.9239)^2} \cdot 0.0081 \right)^2} \quad (47)$$

$$u_R \approx 8.26055 \quad (48)$$

Damit ist das Ergebnis (gerundet):

$$R = (258 \pm 9) \Omega \quad (49)$$

Mathematica Code 6: Unkorrelierte Fehlerfortpflanzung

Wir können es natürlich mit Mathematicas symbolische Berechnungen manuell machen. Dabei können wir die Ableitung durch den Befehl **D** berechnen lassen:

```
rr=uu/ii;
ur=Sqrt[(D[rr,uu]*uU)^2+(D[rr,ii]*uI)^2];
{rr,ur}/.{uu->238.46,uU->7.34,ii->0.9239,uI->0.0081}
```

```
{258.10153, 8.2605547}
```

Seit Mathematica v12.3 allerdings, können wir die automatische Fehlerfortpflanzung mit **Around** benutzen. Alternativ kann man auch **AroundReplace** dazu benutzen:


```
u = Around[238.46, 7.34];
i = Around[0.9239, 0.0081];
u/i
AroundReplace[uu/ii,{uu->u,ii->i}]
```

```
258±8
```

```
258±8
```

7.2 Exkurs: Korrelation

Ohne viel Hintergrundwissen gilt allgemein: zwei Größen sind miteinander Korreliert, wenn man die eine nicht ohne die andere wählen kann. Man kann sich dieses Prinzip sehr leicht an einer



linearen Funktion erklären: In der nebenstehenden Abbildung wurde eine lineare Funktion $f(x) = a \cdot x + b$ möglichst passend durch die vorhandenen Punkte gelegt (durchgezogene Linie; wir werden das im nächsten Kapitel unter dem Namen Regression kennenlernen). Wenn man sich nun vorstellt, dass man den Anstieg a dieser Gerade nach oben zwingt (gepunktete Linie), dann muss der Achsenabschnitt b sinken damit die Gerade wieder relativ gut in den Punkten liegt (gestrichelte Linie). Das gleiche gilt auch

andersherum, zwingt man den Anstieg nach unten, steigt der Achsenabschnitt. Natürlich gilt das auch in vertauschten Rollen, also zwingt man den Achsenabschnitt zu sinken, muss der Anstieg steigen und andersherum. Man merkt damit, dass man die beiden Parameter der linearen Funktion nicht unabhängig voneinander wählen kann. Sie sind also korreliert. Dieses Konzept wird für Fehlerfortpflanzung wichtig, da man für jede Korrelation einen weiteren Term bekommt

der berechnet werden muss.

Um diese Korrelation zwischen zwei Größen x_i und x_j zu quantifizieren führt man eine Größe ein, die man ganz einfach Korrelationskoeffizient $C_{i,j}$ nennt. Dieser kann Werte von $+1$ (volle Korrelation, die Größen ändern sich miteinander in eine Richtung), bis -1 (vollständige Antikorrelation, die Größen ändern sich gegeneinander in die entgegengesetzte Richtung) annehmen. Ein Wert von $C_{i,j} = 0$ bedeutet die Größen sind unkorreliert und können sich unabhängig voneinander ändern. Zugehörig zum Korrelationskoeffizient $C_{i,j}$ gibt es die Kovarianz $u_{i,j}$. Diese Kovarianz ist in gewisser Weise sehr ähnlich zur Varianz u_i^2 einer einzigen Variable. Alle diese Größen kann man auch zusammen in einer Formel darstellen:

Definition 11: Zusammenhang Korrelation, Kovarianz und Unsicherheit

$$C_{i,j} = \frac{u_{i,j}}{u_i \cdot u_j} \quad (50)$$

- $C_{i,j}$ Korrelationskoeffizient $C_{i,j} \in [-1, +1]$ der Werte x_i und x_j
- $u_{i,j}$ Kovarianz der Werte x_i und x_j
- u_i i -te Unsicherheit des x_i Werts
- u_j j -te Unsicherheit des x_j Werts

Einige nützliche Eigenschaften lassen sich aus dieser Formel ablesen. Zum einen, sind die Korrelationsgrößen immer symmetrisch, d.h. $C_{i,j} = C_{j,i}$ sowie $u_{i,j} = u_{j,i}$. Des Weiteren kann die Kovarianz im Betrag nie größer werden als die Multiplikation der dazugehörigen Unsicherheiten $|u_{i,j}| \leq u_i \cdot u_j$.

Jetzt stellt sich natürlich die Frage, wie berechnen wir denn die Korrelation oder Kovarianz? Eine Möglichkeit besteht, wenn man beide Größen stationär mehrmals gemessen hat. Wie auch mit der empirischen Standardabweichung gibt es eine empirische Formel für den Korrelationskoeffizient. Der sogenannte Pearson Korrelationskoeffizient:

$$C_{i,j} = \frac{1}{(n-1) \cdot \sigma_i \cdot \sigma_j} \sum_k^n (x_{i,k} - \bar{x}_i) \cdot (x_{j,k} - \bar{x}_j) \quad (51)$$

Wobei σ_i die Standardabweichung von der Messreihe x_i ist, $x_{i,k}$ das k -te Element von x_i und \bar{x}_i der Mittelwert der Messreihe. Man kann diesen auch als reine Summe darstellen:

$$C_{i,j} = \frac{\sum_k^n (x_{i,k} - \bar{x}_i) \cdot (x_{j,k} - \bar{x}_j)}{\sqrt{\sum_k^n (x_{i,k} - \bar{x}_i)^2} \cdot \sqrt{\sum_k^n (x_{j,k} - \bar{x}_j)^2}} \quad (52)$$

Hat man diesen Wert ausgerechnet, kann man leicht über die Unsicherheiten dann die Kovarianz abschätzen.

Eine zweite Methode die Kovarianzen und Korrelationen abzuschätzen werden wir später im Kapitel zu Regressionen noch kennenlernen.

Mathematica Code 7: Pearson Korrelationskoeffizient

Mathematica kennt den Befehl **Correlation** für den Korrelationskoeffizient:

```
x1={-3.49,2.33,0.63,2.8,-4.72,-1.84,1.81,0.36,-1.99,-0.65};
x2={0.78,-0.53,-0.28,-0.39,0.75,0.60,-0.45,0.05,0.42,0.12};
Correlation[x1,x2]
```

```
-0.961665
```

7.3 Korrelierte Fortpflanzung

Für jede Korrelation der beiden Variablen x_i und x_j mit der zugehörigen Kovarianz $u_{i,j}$, addiert sich ein weiterer Term der Form

$$2 \cdot \frac{\partial f(x_1, x_2, \dots, x_n)}{\partial x_i} \cdot \frac{\partial f(x_1, x_2, \dots, x_n)}{\partial x_j} \cdot u_{i,j} \quad (53)$$

mit unter die Wurzel der unkorrelierten Fortpflanzung. Man kann also allgemein schreiben:

Definition 12: Gaußsche Fehlerfortpflanzung

$$u_f = \sqrt{\sum_i^n \left(\frac{\partial f(x_1, \dots)}{\partial x_i} \cdot u_i \right)^2 + 2 \sum_i^{n-1} \sum_{j=i+1}^n \frac{\partial f(x_1, \dots)}{\partial x_i} \cdot \frac{\partial f(x_1, \dots)}{\partial x_j} \cdot u_{i,j}} \quad (54)$$

- u_f Fortgepflanzte Unsicherheit der berechneten Funktion $f(x_1, x_2, \dots, x_n)$
- n Anzahl aller Werte
- x_i i -te Variable/Wert
- u_i i -te Unsicherheit
- $u_{i,j}$ Kovarianz der Werte x_i und x_j

Beispiel 8: Korrelierte Fehlerfortpflanzung

Gleiches Beispiel wie zuvor im unkorrelierten Beispiel mit den selben Werten und einem Kovarianzfaktor von

$$u_{U,I} = -0.0545 \quad (55)$$

Es folgt mit Korrelationsterm:

$$u_R = \sqrt{\left(\frac{\partial R}{\partial U} \cdot u_U\right)^2 + \left(\frac{\partial R}{\partial I} \cdot u_I\right)^2 + 2 \cdot \frac{\partial R}{\partial U} \cdot \frac{\partial R}{\partial I} \cdot u_{U,I}} \quad (56)$$

$$= \sqrt{\left(\frac{1}{I} \cdot u_U\right)^2 + \left(-\frac{U}{I^2} \cdot u_I\right)^2 - 2 \cdot \frac{U}{I^3} \cdot u_{U,I}} \quad (57)$$

$$= \sqrt{\left(\frac{1}{0.9239} \cdot 7.34\right)^2 + \left(\frac{238.46}{(0.9239)^2} \cdot 0.0081\right)^2 - 2 \cdot \frac{238.46}{(0.9239)^3} (-0.0545)} \quad (58)$$

$$u_R \approx 10.0596 \quad (59)$$

Und somit das Ergebnis (gerundet):

$$R = (258 \pm 10) \Omega \quad (60)$$

Wichtige Anmerkung 3

Korrelationsterme können (anders als unkorrelierte) die Unsicherheit sowohl in positive als auch negative Richtung verändern.

Wenn man sich die Formel für die korrelierte Fehlerfortpflanzung so ansieht, könnte man darauf kommen, dass dahinter eine Matrix-Vektor Multiplikation steht. Tatsächlich, wenn man eine Kovarianzmatrix \underline{U} definiert in der Form $(\underline{U})_{i,j} = u_{i,j}$ dann kann man schreiben:

$$u_f = \sqrt{\nabla^T f \cdot \underline{U} \cdot \nabla f} \quad (61)$$

Diese Matrix \underline{U} wird uns im Kapitel über Regressionen begegnen.

Mathematica Code 8: Korrelierte Fehlerfortpflanzung

Manuell können wir wieder mühevoll eintippen:

```
rr=uu/ii;
ur=Sqrt[(D[rr,uu]*uU)^2+(D[rr,ii]*uI)^2+2*D[rr,uu]*D[rr,ii]*uui];
{rr,ur}/.{uu->238.46,uU->7.34,ii->0.9239,uI->0.0081,uui->-0.0545}

{258.10153,10.059585}
```

However, wir wollen ja eh lieber die automatische Fehlerfortpflanzung. Korreliert können wir diese mit dem Befehl **VectorAround** angeben. Dieser will die Mittelwerte sowie die Kovarianzmatrix. Anwenden können wir das ganze in **AroundReplace**.

```
ui=VectorAround[{238.46,0.9239},{7.34^2,-0.0545},{-0.0545,0.0081^2}]
AroundReplace[ui/ii,{uu,ii}->ui]
```

258±10

8 Regressionen

Es gibt viele Wege wie man die Regression (auch: Ausgleichsrechnung oder Fit) fachlich und mathematisch richtig einführen kann. Die einfachste Form ist aber zu sagen: Wir haben n viele Wertepaare (x_i, y_i) mit den dazugehörigen Unsicherheiten $(u_{x,i}, u_{y,i})$ und eine Gleichung $y = f(x, \{\alpha, \beta, \gamma, \dots\})$ die den Verlauf dieser Punkte beschreiben sollte. Wie müssen wir jetzt die Parameter $\alpha, \beta, \gamma, \dots$ dieser Gleichung wählen, dass sie am besten zu unseren Daten passt? Mathematisch basiert das auf einem Verfahren, dass den Abstand zwischen Funktionswerten $y = f(x_i)$ zu gemessenen Werten y_i minimiert.

$$\{\alpha, \beta, \gamma, \dots\} = \min_{\alpha, \beta, \gamma, \dots} \left[\sum_i^n (y_i - f(x_i, \{\alpha, \beta, \gamma, \dots\}))^2 \right] \quad (62)$$

Das genaue Verfahren ist (besonders für nichtlineare Fits) recht umfangreich und soll kein Gegenstand dieser Einführung hier werden. Für gewöhnlich übernehmen Programme/Programmiersprachen diese Arbeit für euch und die Ergebnisse von Regressionen werden dann visuell dargestellt.

8.1 Fit-Modell und Linearität

Wenn es um Regressionen geht, gibt es grundlegend zwei verschiedene Typen: lineare Regressionen und nichtlineare Regressionen. Eine lineare Regression muss nicht zwingend eine lineare Funktion erzeugen. Eine Regression ist dann linear, wenn die Funktion nach der sie gemacht wird ausgedrückt werden kann als die alleinigen Regressionsparameter $\alpha, \beta, \gamma, \dots$ welche linear




Abbildung 2: Messung und Regression einer ballistischen Flugkurve $h(t) = -g/2 \cdot t^2 + v_0 \cdot t + s_0$. Die Parameter g, v_0, s_0 wurden durch Regression mit Unsicherheiten aus den Daten bestimmt.

kombiniert werden mit linear unabhängigen Basisfunktionen $f_1(x), f_2(x), f_3(x), \dots$

$$y = f(x) = \alpha \cdot f_1(x) + \beta \cdot f_2(x) + \gamma \cdot f_3(x) + \dots = \sum_i p_i \cdot f_i(x) \quad (63)$$

Beispiel 9: Unterschied nicht/-lineare Regressionen

Funktion	Linear/Nichtlinear
$f(x) = a \cdot x + b$	Linear
$f(x) = a^2 \cdot x + b$	Nichtlinear weil a nichtlinear
$f(x) = a \cdot e^x + b \cdot \log x$	Linear weil Linearkombination aus a, b und $e^x, \log x$
$f(x) = \sin(a \cdot x)$	Nichtlinear weil keine Linearkombination
$f(x) = a \cdot x + 2 \cdot b \cdot x$	Nichtlinear weil Basisfunktionen $x, 2x$ nicht unabhängig
$f(x) = a \cdot e^{-x} + b \cdot x + b \cdot \pi$	Nichtlinear weil b doppelt vorhanden
$f(x) = a \cdot e^{-x} + b \cdot (x + \pi)$	Linear linearisiertes Beispiel von Zeile davor
$f(x) = a(x+1)^2 + b(x+2)^2$	Linear weil $(x+1)^2$ und $(x+2)^2$ linear unabhängig
$f(x) = a(x+1) + b \cdot x + c$	Nichtlinear da $(x+1)$ und $x, 1$ linear abhängig

Wichtige Anmerkung 4

Ein Regressionsmodell kann auch über mehrere Basisfunktionen nichtlinear werden. Beispielsweise war das Modell $f(x) = a(x+1)^2 + b(x+2)^2$ linear, da die Basisfunktionen $(x+1)^2$ und $(x+2)^2$ linear unabhängig zueinander sind. Das Modell $f(x) = a(x+1) + b(x+2)$ ist somit ebenfalls linear. Aber das Modell $f(x) = a(x+1) + b(x+2) + c(x+3)$ ist nichtlinear da eine der drei Basisfunktionen immer als Linearkombination der anderen beiden geschrieben werden kann. Sowas kommt fairerweise sehr selten vor.

Mathematica Code 9: Lineare und nichtlineare Regression

Die nichtlineare Regression in Mathematica findet man unter dem Befehl **NonlinearModelFit**. Sie gibt einen `FittedModelObject` zurück, welche man nach verschiedensten Eigenschaften fragen kann. In unserem Fall nach den Werten der angepassten Parameter, deren Unsicherheiten und eine Zusammenfassung der Selbigen.

```
data={{-2,-0.62},{-1.5,0.35},{-1,0.98},{-0.5,1.4},{0,1.7},{0.5,1.91},
      {1,2.09},{1.5,2.24},{2,2.37}};
fit=NonlinearModelFit[data,a*Exp[-x]+b*x+c,{a,b,c},x];
fit["BestFitParameters"] (*Werte*)
fit["ParameterErrors"] (*Unsicherheiten*)
fit["ParameterTable"] (*Zusammenfassung*)
```

```
{a -> -0.297539, b -> 0.207231, c -> 1.99394}
{0.00111616, 0.00199756, 0.00259653}
```

	Estimate	Standard Error	t-Statistics	P-Value
<i>a</i>	-0.29753859	0.0011161648	-266.57227	$1.8806922 \cdot 10^{-13}$
<i>b</i>	0.20723109	0.0019975644	103.74188	$5.4068414 \cdot 10^{-11}$
<i>c</i>	1.9939419	0.0025965338	767.9245	$3.2914029 \cdot 10^{-16}$

Die lineare Regression findet man einfach unter den Namen **LinearModelFit**. Diese Funktion will keine Formel mit Parametern, sondern nur eine Liste der Basisfunktionen. Man kann auch hier wieder nach allen Eigenschaften von zuvor fragen. Hier diesmal nur die Werte und Unsicherheiten.

```
data={{-2,-0.62},{-1.5,0.35},{-1,0.98},{-0.5,1.4},{0,1.7},{0.5,1.91},
      {1,2.09},{1.5,2.24},{2,2.37}};
fit=LinearModelFit[data,{Exp[-x],x,1},x];
fit["BestFitParameters"] (*Werte*)
fit["ParameterErrors"] (*Unsicherheiten*)
```

```
{-0.297539, 0.207231, 1.99394}
{0.00111616, 0.00199756, 0.00259653}
```

8.2 Gewichtete Regression

Wie auch beim gewichteten Mittel haben wir in der Regression die Option genauere/präzisere Messpunkte stärker in das Ergebnis einfließen zu lassen. Auch hier gilt die bereits bekannte Instrumentelle Gewichtung, d.h. zu jedem Datenpunkt (x_i, y_i) mit den Unsicherheiten $(u_{x,i}, u_{y,i})$ existiert die Gewichtung p_i mit

$$p_i = \frac{1}{u_{y,i}^2} \quad (64)$$

Hierbei fällt auf, dass nur die Unsicherheit von y_i eine Rolle spielt. Dies macht man mit der Näherung, dass $u_{x,i} \ll u_{y,i}$. Sollte diese Näherung umgekehrt gegeben sein, d.h. $u_{x,i} \gg u_{y,i}$, kann man mit $y = f(x)$ die Umkehrfunktion bilden $x = f^{-1}(y)$ und nach (y_i, x_i) mit den Gewichtungen $p_i = 1/u_{x,i}^2$ fitten.

Wichtige Anmerkung 5

Sollte weder die Bedingung $u_{x,i} \ll u_{y,i}$ noch $u_{x,i} \gg u_{y,i}$ erfüllt sein, dann muss man theoretisch eine sogenannte Orthogonalregression machen. Diese ist aber furchtbar kompliziert und nicht Gegenstand dieses Scripts.

Mathematica Code 10: Gewichtete Regression

Mit den Daten und Modell von zuvor kann man gewichten, indem man die **Weights** Option benutzt.

```
unsicherheiten={-2.18,-1.28,-0.71,-0.32,0.1,0.3,0.71,1.28,2.19};
gewichte=1/unsicherheiten^2;
fit=NonlinearModelFit[data,a*Exp[-x]+b*x+c,{a,b,c},x,Weights->gewichte];
fit["BestFitParameters"]
```

```
{a -> -0.3022283, b -> 0.19789626, c -> 2.0011934}
```

8.3 R^2 -Test

Der R^2 -Test oder auch Bestimmtheitsmaß genannt, gibt an wie viel Prozent der Abweichung von dem Fitmodell erklärt werden kann über aufgespannte Flächen der Residuen. Es ist definiert mit:

$$R^2 = 1 - \frac{\sum_i^n (f(x_i) - y_i)^2}{\sum_i^n (\bar{y} - y_i)^2} \quad (65)$$

und kann Werte von null bis eins annehmen. Dabei gilt: je näher der Wert an eins, desto besser. Übliche/gute Werte im Grundpraktikum kommen auf 0.99 und mehr und dieser Test ist in der Physik auch von geringerer Bedeutung.

Beim normalen R^2 -Test gibt es ein Problem. Steigt die Anzahl der Parameter, dann steigt zwangsläufig auch der R^2 -Test mit an. Um den entgegen zu wirken, kann ein korrigierter Test angegeben werden mit p als die Anzahl der Parameter im Fitmodell:

$$\bar{R}^2 = R^2 - (1 - R^2) \cdot \frac{p}{n - p - 1} \quad (66)$$

Wichtige Anmerkung 6

Der R^2 -Test ist in der Physik von eher geringerer Bedeutung, da bei uns die Messungen eher weniger stark streuen und somit übliche Werte von Anfang an $R^2 \approx 1$ sind.

Der Test eignet sich auch wirklich nur bei linearen Funktionen als Fitmodell.

Zuletzt muss gesagt werden, dass der Test nicht angibt ob ein Fitmodell gut zu den Daten passt oder konvergiert ist.

Mathematica Code 11: R^2 -Test

Hat man einen Fit gemacht, kann man von diesem den R^2 -Test und \overline{R}^2 -Test als Eigenschaft abfragen:

```
fit=LinearModelFit[data, {Exp[-x], x, 1}, x];
fit["RSquared"]
fit["AdjustedRSquared"]
```

```
0.99999008
```

```
0.99998678
```

8.4 χ^2/dof -Test

Der χ^2/dof -Test (*dof* steht für Freiheitsgrade [**d**egrees **o**f **f**reedom]) und kann berechnet werden mit $dof = n - p$ wobei p die Anzahl an Fitparameter angibt) gibt begrenzte Auskunft darüber wie gut der Fit innerhalb der Unsicherheiten an die Daten angepasst ist. Er kann Werte von null bis unendlich annehmen, wobei ein Wert von $n/(n - p) \stackrel{n \gg p}{\approx} 1$ als ein guter Test gilt. Abweichungen von diesem Nominalwert können mehrere Gründe haben:

- $\chi^2/dof \ll 1$: Entweder zu große Unsicherheiten oder relativ zu den Unsicherheiten überdurchschnittlich gute Anpassung des Fits an die Daten.
- $\chi^2/dof \gg 1$: Entweder zu kleine Unsicherheiten oder relativ zu den Unsicherheiten unterdurchschnittlich schlechte Anpassung des Fits an die Daten.

Die tatsächlich, finale Einschätzung unterliegt aber einer leichten Subjektivität. Definiert ist der Test über die eingangs erwähnten Abstandsqadrate der Messpunkte (x_i, y_i) mit der Fitfunktion $y = f(x)$ und den zu den Messpunkten gehörenden Messunsicherheiten $(u_{x,i}, u_{y,i})$:

$$\frac{\chi^2}{dof} = \frac{1}{n - p} \sum_i^n \frac{(f(x_i) - y_i)^2}{u_{y,i}^2} \quad (67)$$

Mathematica Code 12: χ^2/dof -Test

You can get the test either by looking in the ANOVA Table (**A**nalysis of **V**ariance Table - Error Zeile, MS Spalte) oder man holt sich das direkt programmatisch:

```
fit=LinearModelFit[data,{Exp[-x],x,1},x];
fit["ANOVATable"]
Last[fit["ANOVATableMeanSquares"]]
```

	DF	SS	MS	F-Statistics	P-Value
x	1	0.19775531	0.19775531	15282.992	$1.8889965 \times 10^{-11}$
1	1	7.6305671	7.6305671	589708.03	$3.2914029 \times 10^{-16}$
Error	6	0.000077637407	0.000012939568		
Total	8	7.8284			

0.000012939568

8.5 Kovarianz- und Korrelationsmatrix

Aus dem Kapitel zur korrelierten Gaußschen Fehlerfortpflanzung wissen wir, dass wir dazu einen Kovarianzfaktor $u_{i,j}$ brauchen. Abgesehen von einer Abschätzung per Hand durch $u_{i,j} = C_{i,j} \cdot u_i \cdot u_j$ oder Messung und Berechnung von Formel (51), können diese Werte aus einer Regression in Form einer Kovarianzmatrix gewonnen werden. Diese hat auf den Diagonalen die Quadrate der Unsicherheiten der Parameter und außerhalb die jeweiligen, gesuchten Kovarianzen. Dazu parallel, kann man natürlich auch eine Korrelationsmatrix angeben

Definition 13: Kovarianzmatrix und Korrelationsmatrix

$$\text{cov}(\alpha, \beta, \gamma, \dots) = \begin{pmatrix} u_\alpha^2 & u_{\alpha,\beta} & u_{\alpha,\gamma} & \dots \\ u_{\beta,\alpha} & u_\beta^2 & u_{\beta,\gamma} & \dots \\ u_{\gamma,\alpha} & u_{\gamma,\beta} & u_\gamma^2 & \dots \\ \vdots & \vdots & \vdots & \ddots \end{pmatrix} \quad (68)$$

$$\text{corr}(\alpha, \beta, \gamma, \dots) = \begin{pmatrix} 1 & C_{\alpha,\beta} & C_{\alpha,\gamma} & \dots \\ C_{\beta,\alpha} & 1 & C_{\beta,\gamma} & \dots \\ C_{\gamma,\alpha} & C_{\gamma,\beta} & 1 & \dots \\ \vdots & \vdots & \vdots & \ddots \end{pmatrix} \quad (69)$$

$\alpha, \beta, \gamma, \dots$	Fitparameter
$\text{cov}(\alpha, \beta, \gamma, \dots)$	Kovarianzmatrix
$\text{corr}(\alpha, \beta, \gamma, \dots)$	Korrelationsmatrix
$u_\alpha, u_\beta, u_\gamma, \dots$	Unsicherheiten der Fitparameter
$u_{\alpha,\beta}, \dots$	Kovarianz zwischen α und β
$C_{\gamma,\alpha}, \dots$	Korrelationskoeffizient zwischen α und γ

Mathematica Code 13: Kovarianzmatrix und Korrelationsmatrix

Für beide Matrizen existieren wieder einfache, abfragbare Eigenschaften:

```
fit=LinearModelFit[data,{Exp[-x],x,1},x];
fit["CovarianceMatrix"]
fit["CorrelationMatrix"]
```

$$\begin{pmatrix} 1.245824 \cdot 10^{-6} & 1.9739481 \cdot 10^{-6} & -2.5706365 \cdot 10^{-6} \\ 1.9739481 \cdot 10^{-6} & 3.9902635 \cdot 10^{-6} & -4.0730497 \cdot 10^{-6} \\ -2.5706365 \cdot 10^{-6} & -4.0730497 \cdot 10^{-6} & 6.7419879 \cdot 10^{-6} \end{pmatrix}$$

$$\begin{pmatrix} 1. & 0.88533288 & -0.88698922 \\ 0.88533288 & 1. & -0.78528072 \\ -0.88698922 & -0.78528072 & 1. \end{pmatrix}$$

8.6 Korrelationen linearer Funktionen

Führt man eine Regression über eine lineare Funktion der Form $f(x) = a \cdot x + b$ durch und benutzt beide Parameter a und b später um den Bruch b/a oder a/b zu berechnen, muss man das unter voller Beachtung der Korrelation tun. Es gibt aber einen Trick um sich die gesamte korrelierte Gaußsche Fehlerfortpflanzung zu sparen. Wenn man eine Variable aus der Gleichung ausklammert:

$$f(x) = a \cdot \left(x + \frac{b}{a} \right) \quad (70)$$

$$f(x) = b \cdot \left(\frac{a}{b} \cdot x + 1 \right) \quad (71)$$

und den inneren Bruch-Term mit einem Parameter substituiert

$$f(x) = a \cdot (x + c) \quad (72)$$

$$f(x) = b \cdot (d \cdot x + 1) \quad (73)$$

erhält man mit $c = b/a$ oder $d = a/b$ die Brüche mit kompletter korrelierter Fehlerabschätzung dessen Funktionen nur noch nichtlinear gefittet werden müssen.